3 MATRIKS DAN PERSAMAAN LINIER

Banyak sekali jenis Matriks yang kita kenal. Suatu problema kadang-kadang menjadi lebih mudah apabila kita berhasil membawanya ke dalam suatu model Matriks, dan menyelesaikannya mempergunakan modelMatriks tersebut.
Model tersebut misalnya adalah model m masalah persamaan linier, Invers Matriks ataupun eigenvalue.

3.1 MATRIKS BARIS DAN MATRIKS KOLOM

Misalkan suatu kotak berisi 13 duku, 8 rambutan dan 6 sawo. Banyaknya buah-buahan dalam kotak tersebut dapat kita tulis sebagai A = (13 8 6).
Penulisan ini disebut penulisan dalam *Matriks baris*. Matrika baris disebut juga *vektor baris* berorder 3.
Selain itu, kita dapat menuliskan keadaan di atas sebagai

$$\begin{bmatrix} 13 \\ 8 \\ 6 \end{bmatrix}$$

Bentuk ini disebut *Matriks kolom*, atau *vektor kolom* order3.
Untuk menempatkan Matriks dalam komputer, sebaiknya kita gunakan variabel bersubskrip. Jadi, misalnya pada contoh di atas, kita menyebut A(1) = 13, A(2) = 8 , A(3) = 6.

Berikut ini sebuah Contoh Program mengenai Matriks, dalam bahasa BASIC :

Program 3.1

```
10   REM MEMASUKKAN DAN MENCETAK MATRIKS BARIS
20   INPUT "BANYAKNYA ELEMEN ";N
```

```
 30  DIM A(N)
 40  FOR J = 1 TO N
 50  PRINT "ELEMEN KE ";J;: INPUT A(J)
 60  NEXT J
 70  FOR K = 1 TO N
 80  PRINT TAB (2*K); A(K);
 90  NEXT K
100  END
```

Silakan dicoba, bagaimana kita mencetak Matriks kolom.

Di dalam Aljabar Vektor, kita mengenal beberapa operasi, di antaranya :

Penjumlahan 2 vektor
Misalnya A = [4 3 -5], B = [2 8 3], maka
A + B = [4+2 3+8 -5+3] = [6 11 -2]

Perkalian dengan Skalar, yakni mengalikan vektor dengan suatu bilangan real.
Misalnya A = [5 6 4], maka -3A = [-3*5 -3*6 - 3*4] = [-15 -18 -12]

SOAL LATIHAN : Coba anda buat program untuk mencetak hasil operasi pA + qB, dengan p dan q merupakan bilangan real, A serta B adalah vektor-vektor baris yang diketahui.

Dot produk antara 2 vektor A = [A(1) A(2) ... A(N)], dan vektor B = [B(1) B(2) ... B(N), adalah A(1)*B(1) + A(2)*B(2) + ,,, + A(N)*B(N).
Misalnya A = [4 3 -5], B = [2 8 3]
maka dot produk adalah 4*2 + 3*8 + -5*3 = 8 + 24 -15 = 17

Juga didefinisikan *Norma* atau *Panjang vektor* A sebagai *akar, atau SQR, dari A dot A*, dan 2 vektor dikatakan saling orthogonal (tegak lurus), bila dot produk mereka adalah 0.

Program berikut ini menunjukkan bagaimana memeriksa 2 vektor, apakah saling tegak lurus atau orthogonal.

Program 3.2

```
 10  REM  VEKTOR ORTHOGONAL
 20  INPUT "ORDER VEKTOR ";N
 30  DIM A(N), B(N)
 40  FOR J = 1 TO N
 50  INPUT A(J), B(J)
 60  NEXT J
 70  D = 0
 80  FOR K = 1 TO N
 90  D = D + A(K) * B(K)
100  NEXT K
110  IF D <> O THEN 140
120  PRINT "KEDUA VEKTOR SALING ORTHOGONAL "
130  GO TO 150
140  PRINT "KEDUA VEKTOR TIDAK SALING ORTHOGONAL"
150  END
```

SOAL LATIHAN : Coba anda buat sebuah program untuk menghitung Norma vektor. Sesudah itu kembangkan Program anda tersebut, sehingga mampu menghitung keliling sebuah segi tiga yang diketahui 2 buah vektor pembentuknya.

3.2 MATRIKS (M x N)

Secara umum Matriks terdiri atas M baris dan N kolom. Misalnya Matriks A = 3 1 2

5 -3 4

terdiri dari 2 baris dan 3 kolom. Ia kita sebut juga Matriks berukuran (2 x 3). Kalau banyaknya baris dan kolom sama, maka dikatakan bahwa Matriks merupakan *Matriks bujur sangkar*.

Untuk menyatakan suatu Matriks ukuran (M x N), kita gunakan variabel bersubskrip. Untuk Matriks A di atas, elemen-elemennya dapat kita tulis sebagai :

A(1,1) = 3 , A(1,2) = 1 , A(1,3) = 2
A(2,2) = 5 , A(2,2) =-3 , A(2,3) = 4

Berikut ini program untuk menentukan dan mencetak Matriks berukuran (M x N).

Program 3.3

```
 10  REM PROGRAM MATRIKS M KALI N
 15  INPUT "BANYAK BARIS"; M
 20  INPUT "BANYAK KOLOM"; N
 30  DIM A(M,N)
 40  PRINT "MASUKKAN ELEMEN BARIS DEMI BARIS"
 50  FOR I = 1 TO M
 60      FOR J = 1 TO N
 70      INPUT A(I,J);
 80      NEXT J
 90  PRINT
100  NEXT I
110  REM MENCETAK MATRIKS
120  FOR I = 1 TO M
130      FOR J = 1 TO N
140      PRINT TAB (2*J); A(I,J);
150      NEXT J
160  PRINT
170  NEXT I
180  END
```

Pertanyaan: Apa kiranya kegunaan PRINT pada baris 160 ?

Seperti halnya pada vektor, kita mengenal penjumlahan dua Matriks dan perkalian Matriks dengan skalar.

Misalnya Matriks A = $\begin{bmatrix} 4 & 2 & 1 \\ 3 & 0 & 2 \end{bmatrix}$ dan B = $\begin{bmatrix} 6 & -4 & 8 \\ 2 & 8 & 3 \end{bmatrix}$

Maka $A + B = \begin{bmatrix} 4+6 & 2-4 & 1+8 \\ 3+2 & 0+8 & 2+3 \end{bmatrix} = \begin{bmatrix} 10 & -2 & 9 \\ 5 & 8 & 5 \end{bmatrix}$

dan $-2B = \begin{bmatrix} -2*6 & -2*-4 & -2*8 \\ 2*2 & -2*8 & -2*3 \end{bmatrix} = \begin{bmatrix} -12 & 8 & -16 \\ -4 & -16 & -6 \end{bmatrix}$

Berikut ini program untuk menghitung pA + pQ, di mana p dan q bilangan, A dan B dan Matriks berukuran

Program 3.4

```
 10  REM PROGRAM PENJUMLAHAN DAN PERKALIAN SKALAR
 15  INPUT "BANYAK BARIS ";M
 20  INPUT "BANYAK KOLOM ";N
 30  DIM A(M,N), B(M,N), C(M,N)
 40  INPUT P,Q
 50  FOR I =1 TO M
 60      FOR J = 1 TO M : INPUT A(I,J), B(I,J) : NEXT J
 70  NEXT I
 80  FOR I = 1 TO M
 90      FOR J = 1 TO N
100      C(I,J) = P * A(I,J) + Q * B(I,J)
110      NEXT J
120  NEXT I
130  REM MENCETAK HASILNYA
140  FOR I = 1 TO M
150      FOR J = 1 TO N : PRINT TAB (2*J); C(I,J); : NEXT J
160  PRINT
170  NEXT I
180  END
```

3.3 PERKALIAN MATRIKS

Perkalian Matriks hanya terdefinisi bila *banyak kolom Matriks pertama = banyak baris Matriks kedua.*

Misalnya Matriks $A = \begin{bmatrix} 3 & 1 & 1 \\ 0 & 2 & 1 \end{bmatrix}$; $B = \begin{bmatrix} 1 & 1 & 1 \\ 0 & 2 & 1 \\ 1 & 3 & 0 \end{bmatrix}$

Ukuran A adalah (2 x 3), dan ukuran B (3 x 3), maka Matriks hasilkalinya , AB terdefinisi. Matriks hasil tersebut, sebut sebagai C, berukuran (2 x 3). Sedangkan hasilkali BA tidak terdefinisi.

Perhatikan bagaimana kita mengalikan A dan B

$$\begin{bmatrix} 3 & 1 & 1 \\ 0 & 2 & 1 \end{bmatrix} \begin{bmatrix} 1 & 1 & 1 \\ 0 & 2 & 1 \\ 1 & 3 & 0 \end{bmatrix} = \begin{bmatrix} 3*1+1*0+1*1 & \ldots & \ldots \\ .. & \ldots & 0*1+2*1+1*0 \end{bmatrix}$$

Di sini

C(1,1) = A(1,1)*B(1,1)+A(1,2)*B(2,1)+A(1,3)*B(3,1)

....

C(2,3) = A(2,1)*B(1,3)+A(2,2)*B(2,3)+A(2,3)*B(3,3)

atau secara umum

C(I,J) = A(I,1)*B(1,J)+A(I,2)*B(2,J)+A(I,3)*B(3,J)

(di sini terjadi 3 suku jumlahan, karena banyak kolom A = banyak baris B = 3).

Berikut ini programnya:

Program 3.5.

```
10   REM PROGRAM PERKALIAN MATRIKS
20   INPUT "BANYAK BARIS, BANYAK KOLOM MATRIKS #1";P,Q
25   INPUT "BANYAK BARIS, BANYAK KOLOM MATRIKS #2";R,S
30   IF Q <> R THEN PRINT "TIDAK TERDEFINISI" : STOP
40   DIM A(P,Q), B(R,S), C(P,S)
50   FOR I = 1 TO P : FOR J = 1 TO Q : INPUT A(I,J) : NEXT J,I
60   FOR I = 1 TO R : FOR J = 1 TO S : INPUT B(I,J) : NEXT J,I
70   FOR I = 1 TO P
80        FOR J = 1 TO S
90        C(I,J) = 0
100          FOR K = 1 TO Q
110          C(I,J) = C(I,J) + A(I,K) * B(K,J)
120          NEXT K
130       NEXT J
140  NEXT I
150  FOR I = 1 TO P
160       FOR J = 1 TO S : PRINT TAB(2*J) : C(I,J); : NEXT J
170  PRINT
180  NEXT I
190  END
```

3.4 TRANSPOSE MATRIKS

Apabila penulisan baris kita ubah menjadi penulisan kolom, dikatakan bahwa kita melakukan *transpose* terhadap Matriks tersebut.

Misalnya A = $\begin{bmatrix} 2 & 1 & 1 \\ 3 & 0 & 6 \end{bmatrix}$ kita tulis menjadi $\begin{bmatrix} 2 & 3 \\ 1 & 0 \\ 1 & 6 \end{bmatrix}$

Kita notasikan transpose dari Matriks A sebagai A^T. Jelas bahwa setiap elemen (I,J) dari Matriks A akan menjadi elemen (J,I) dari transposenya, sehingga bila ukuran Matriks A adalah (M x N), maka ukurantransposenya, A^T adalah (N x M)

Program 3.6

```
10   REM PROGRAM TRANSPOSE MATRIKS
20   INPUT "UKURAN MATRIKS ";M,N
30   DIM A(M.N), T(N,M)
40   FOR I = 1 TO M : FOR J = 1 TO N : INPUT A(I,J) : NEXT I
50   FOR I = 1 TO M
60        FOR J = 1 TO N
70        T(J,I) = A(I,J)
80        NEXT J
85   REM MENCETAK TRANSPOSE
90   NEXT I
100  FOR I = 1 TO N : FOR J = 1 TO M : PRINT TAB(2*J) ;T(I,J);NEXT J : PRINT :
     NEXT i
110  END
```

MATRIKS SIMETRIS DAN ANTI SIMETRIS

Suatu Matriks bujur sangkar disebut Matriks Simetris bila $A = A^T$, dan disebut Anti Simetris bila $A = -A^T$.

Contoh Matriks Simetris $\begin{bmatrix} 3 & 6 \\ 6 & 5 \end{bmatrix}$ dan contoh Matriks Anti Simetris $\begin{bmatrix} 0 & -6 \\ 6 & 0 \end{bmatrix}$

SOAL LATIHAN :Silakan dibuat program untuk memeriksa apakah suatu Matriks Simetris, dan program untuk memeriksa apakah suatu Matriks Anti Simetris.

3.5 DETERMINAN

Untuk Matriks ukuran (2 x 2)

$$A = \begin{bmatrix} A(1,1) & A(1,2) \\ A(2,1) & A(2,2) \end{bmatrix}$$

kita kenal determinan dari A adalah

DET(A)=A(1,1)*A(2,2) - A(1,2)*A(2,1).

Sebagai contoh :

$$\det \begin{bmatrix} 3 & 5 \\ 4 & 2 \end{bmatrix}$$

adalah 3*2 - 5*4 = -14

Sedangkan untuk Matriks (3 x 3); kita dapat menggunakan cara Sarus sebagai berikut :

$$\text{Determinan} \begin{bmatrix} A(1,1) & A(1,2) & A(1,3) \\ A(2,1) & A(2,2) & A(2,3) \\ A(3,1) & A(3,2) & A(3,3) \end{bmatrix}$$

= A(1,1) * A(2,2) * A(3,3) + A(1,2) * A(2,3) * A(3,1)
+ A(1,3) * A(2,1) * A(3,2) - A(1,3) * A(2,2) * A(3,1)
- A(1,2) * A(2,1) * A(3,3) - A(1,1) * A(2,3) * A(3,2)

Sebagai contoh adalah :

$$\text{Det} \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 1 \end{bmatrix} = 24$$

LATIHAN SOAL : Coba buat program untuk menghitung determinan Matriks (2 x 2) serta Matriks (3 x 3).

Secara umum kita dapat menghitung determinan dengan *Metode Maksimum Pivot..* Ikutilah algoritma metode tersebut dengan contoh berikut ini. Misal

$$A = \begin{bmatrix} 2 & 1 & 4 \\ 1 & 4 & 8 \\ 4 & 8 & 2 \end{bmatrix}$$

(1) Mula-mula kita misalkan harga awal determinan = D = 1

(2) Cari elemen terbesar dari Matriks, dalam contoh kita ini adalah A(2,3) = 8

(3) Kitalakukan pertukaran baris serta kolom (transformasi elementer baris serta kolom) sedemikian sehingga elemen 8 menjadi elemen A(1,1)

Pertukaran basis 1 dan 2 $\begin{bmatrix} 1 & 4 & 8 \\ 2 & 1 & 4 \\ 4 & 8 & 2 \end{bmatrix}$

Kita ingat salah satu sifat determinan, yaitu nilai determinan hanya berubah tanda, bila dilakukan satu kali pertukaran baris, atau satu kali pertukaran kolom.

Jadi sekarang D = -1

Dilakukan pertukaran kolom 1 dan 3 $\begin{bmatrix} 8 & 4 & 1 \\ 4 & 1 & 2 \\ 2 & 8 & 4 \end{bmatrix}$

D kembali = 1

(4) Elemen A(1,1) = 8 disebut *Pivot.*

Kita bagi baris 1 dengan 8, sebagai berikut : $\begin{bmatrix} 1 & 0.5 & 0.125 \\ 4 & 1 & 2 \\ 2 & 8 & 4 \end{bmatrix}$

Menurut sifat determinan pembagian tersebut menyebabkan nilai determinan Matriks baru adalah seperdelapan nilai determinan yang kita cari. Jadi D = 8.

(5) Kita nolkan elemen yang sekolom pivot, yaitu menolkan elemen A(2,1) dengan cara menghitung ratio R = A(2,1) = 4, alu menghitung

A(2,2) = A(2,2) - R * A(1,2) = 1 - 4 * 0.5 = -1
A(2,3) = A(2,3) - R * A(1,3) = 2 - 4 * 0.125 = 1.5

dan menolkan elemen A(3,1), dengan ratio R = A(3,1) = 2, lalu menghitung

A(3,2) = A(3,2) - R * A(1,2) = 8 - 2 * 0.5 = 7
A(3,3) =A(3,3) - R * A(1,3) = 4 - 2 * 0.125 = 3.75

Sehingga diperoleh $\begin{bmatrix} 1 & 0.5 & 0.125 \\ 0 & -1 & 1.5 \\ 0 & 7 & 3.75 \end{bmatrix}$

D tetap = 8

(6) Kita reduksikan ukuran Matriks dengan menghapus baris 1 dan kolom 1, diperoleh

$$\begin{bmatrix} -1 & 1.5 \\ 7 & 3.75 \end{bmatrix}$$

kemudian kita kembali ke langkah memilih elemen terbesar, dan seterusnya.
Terlihat elemen terbesar = 7, dengan pertukaran baris 1 dan 2 diperoleh Matriks

$$\begin{bmatrix} 7 & 3.75 \\ -1 & 1.5 \end{bmatrix}$$

Sekarang D = -8

Kita bagi baris 1 dengan 7 diperoleh $\begin{bmatrix} 1 & 0.5357 \\ -1 & 1.5 \end{bmatrix}$

D = 7 * -8 = -56

Dengan menolkan A(2,1) diperoleh : $\begin{bmatrix} 1 & 0.5357 \\ 0 & 2.0357 \end{bmatrix}$

Sehingga diperoleh hasil akhir Determinan D = 2.0357 * -56 = -114

Berikut ini programnya :

Program 3.6

```
10   REM MENGHITUNG DETERMINAN
15   REM DENGAN METODA MAKSIMUM PIVOT
20   INPUT "ORDER MATRIKS ";N
30   DIM A(N,N)
40   PRINT "MASUKKAN ELEMEN MATRIKS"
50   FOR I = 1 TO N : FOR J = 1 TO N : INPUT A(I,J) : NEXT J,I
60   D = 1
70   GOSUB 200
80   GOSUB 300
90   GOSUB 400
100  GOSUB 500
110  GOTO 70
120  PRINT "DETERMINAN = ";D
130  END
200  REM MENCARI ELEMEN TERBESAR
210  REM B = TERBESAR PADA BARIS K, KOLOM L
220  K = 1 : L = 1 : B = A(K,L)
230  FOR I = 1 TO N
240       FOR J = 1 TO N
250       IFABS(A(I,J)) <= ABS(B) THEN 270
260       K = I : L = J : B = A(K,L)
270       NEXT J
280  NEXT I
290  RETURN
300  REM ELEMEN TERBESAR DILETAKKAN DI BARIS  SEBAGAI PIVOT
310  FOR J = 1 TO N
```

```
320  IF K = 1 THEN 340
330  P = A(1,J) : A(1,J) = A(K,J) : A(K,J) = P
340  NEXT J
350  D = -D
360  FOR I = 1 TO N
370  IF L = 1 THEN 390
380  P = A(I,1) : A(I,1) = A(I,L) : A(I,L) = P
390  NEXT I
395  D = -D : RETURN
400  REM MENGNOLKAN ELEMEN SEKOLOM PIVOT
410  FOR J = 1 TO N
420  A(1,J) = A(1,J)/B
425  NEXT J
430  D = D * B
440  FOR I = 2 TO N
450       FOR J = 1 TO N
460       R = A(I,1)
470       A(I,J) = A(I,J) - R * A(1,J)
480       NEXT J
490  NEXT I
495  RETURN
500  REM MEREDUKSI UKURAN MATRIKS
510  IF N = 2 THEN D = D * A(2,2) : GOTO ?
520  N = N-1
530  FOR I = 1 TO N
540       FOR J = 1 TO N
550       A(I,J) = A(I+1,J+1)
560       NEXT J
570  NEXT I
580  RETURN
```

3.6. MATRIKS INVERS

Suatu Matriks bujursangkar A ukuran (N x N) dikatakan mempunyai Invers Matriks A^{-1}, yang berukuran (N x N), bila berlaku $A*A^{-1} = A^{-1}*A = I_N$ (I_N adalah Matriks identitas berukuran N x N).

Tidak setiap Matriks bujursangkar mempunyai Invers. Syarat mempunyai Invers adalah bahwa Determinan Matriks tersebut *tidak sama dengan nol*, atau Matriks merupakan Matriks *Nonsingular*.

Contoh Matriks I_N, untuk N=3 adalah $\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$

SOAL LATIHAN: Buatlah program untuk membentuk Matriks Identitas berukuran (N x N).

Untuk Matriks A berukuran (2x2), A = $\begin{bmatrix} A(1,1) & A(1,2) \\ A(2,1) & A(2,2) \end{bmatrix}$

dengan DET(A) = A(1,1) * A(2,2) - A(1,2) * A(2,1)
kita mempunyai rumus yang sangat mudah untuk menentukan inversnya, yakni A^{-1} =

$$\frac{1}{\det(A)} \begin{vmatrix} A(2,2) & -A(1,2) \\ -A(2,1) & A(1,1) \end{vmatrix}$$

Jika Matriks A = $\begin{bmatrix} 3 & 2 \\ 4 & 5 \end{bmatrix}$ maka A^{-1} = $\frac{1}{7} \begin{vmatrix} 5 & -2 \\ -4 & 3 \end{vmatrix}$

$$= \begin{bmatrix} 5/7 & -2/7 \\ -4/7 & 3/7 \end{bmatrix}$$

LATIHAN SOAL: Silakan dicoba membuat programnya.

Kita dapat menghitung Invers dengan Metode Adjoint. *Matriks Adjoint* dari Matriks A adalah *Transpose dari Matriks Kofaktor* dari A. Kofaktor dari elemen A(I,J) adalah bilangan kof A(I,J)=$(-1)^{I+J}$*determinan submatriks yang diperoleh dari penghapusan baris I dan kolom J.
Sebagai Contoh, untuk Matriks A = $\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 1 & 3 & 2 \end{bmatrix}$

Kofaktor dari A(1,1) = $(-1)^{1+1}$*det $\begin{bmatrix} 5 & 6 \\ 3 & 2 \end{bmatrix}$ = -8

Kofaktor dari A(2,3) = $(-1)^{2+3}$*det $\begin{bmatrix} 1 & 2 \\ 1 & 3 \end{bmatrix}$ = -1

Jadi lengkapnya :
Adjoint (A) = $\begin{bmatrix} \text{kof } A(1,1) & \text{kof } A(2,1) & \text{kof } A(3,1) \\ \text{kof } A(1,2) & \text{kof } A(2,2) & \text{kof } A(3,2) \\ \text{kof } A(1,3) & \text{kof } A(2,3) & \text{kof } A(3,3) \end{bmatrix}$

= $\begin{bmatrix} -8 & 5 & -3 \\ 2 & -1 & 6 \\ 7 & -1 & -3 \end{bmatrix}$ (Perhatikan adanya transpose)

Bila det(A) dihitung, diperoleh = 9.
Kemudian kita gunakan rumus
A^{-1} = Adjoint (A)/Det(A)

diperoleh :

$$A^{-1} = \begin{bmatrix} -8/9 & 5/9 & -3/9 \\ 2/9 & -1/9 & 6/9 \\ 7/9 & -1/9 & -3/9 \end{bmatrix}$$

Berikut ini sub program untuk menentukan Matriks Adjoint :

Program 3.7

```
 95  REM SUBPROGRAM MATRIKS ADJOINT
100   FOR I = 1 TO N : FOR J = 1 TO N
110     FOR K = 1 TO N-1
120             FOR S = 1 TO N
130       IF K < I THEN B(K,S) = A(K,S) : GO TO
140       B(K,S) = A(K + 1,S)
150       NEXT S
160     NEXT K
170     FOR L = 1 TO N-1
180       FOR R = 1 TO N-1
190       IF L < J THEN 210
200       B(R,L) = B(R,L+1)
210       NEXT R
220     NEXT L
225  REM GOSUB MENGHITUNG DETERMINAN
230  GOSUB 1000
240  KOFA (J,I) = (-1)^(I+J)* DET
250  NEXT J,I
260  FOR I = 1 TO N : FOR J = 1 TO N
270  PRINT TAB(2*J);KOFA (I,J); : PRINT : NEXT J,I
280  END
```

LATIHAN SOAL: Buatlah program lengkap untuk mencari Matriks Invers dengan Metode Adjoint

Selain menggunakan Metode Matriks Adjoint tersebut di atas, kita dapat pula menggunakan Metode Eliminasi atau Penyapuan Gauss (yang dimodifikasi menjadi Metode Eliminasi Gauss-Jordan).Prinsip kerjanya adalah pemanfaatan Sifat Transformasi Elementer Baris Matriks.

Kita ikuti contoh berikut, yakni contoh menghitung Invers Matriks A :

$$\begin{bmatrix} 2 & 4 & 4 \\ 3 & 5 & 3 \\ 2 & 1 & 2 \end{bmatrix}$$

Langkah Penyelesaian:

Mula-mula matriks tersebut kita rangkaikan dengan Matriks Identitas berorder yang sesuai.

$$\begin{bmatrix} 2 & 4 & 4 & 1 & 0 & 0 \\ 3 & 5 & 3 & 0 & 1 & 0 \\ 2 & 1 & 2 & 0 & 0 & 1 \end{bmatrix}$$

Kita akan berusaha menjadikan matriks A semula, menjadi Matriks Identitas. Elemen A(1,1) akan menjadi Pivot untuk menolkan A(2,1) dan A(3,1)

Untuk itu, mula-mula kitabagi baris pertama dengan 2, diperoleh :

$$\begin{bmatrix} 1 & 2 & 2 & 1/2 & 0 & 0 \\ 3 & 5 & 3 & 0 & 1 & 0 \\ 2 & 1 & 2 & 0 & 0 & 1 \end{bmatrix}$$

Kita kurangi baris kedua dengan 3 kali baris pertama, dan baris ketiga dengan dua kali baris pertama, diperoleh :

$$\begin{bmatrix} 1 & 2 & 2 & 1/2 & 0 & 0 \\ 0 & -1 & -3 & -3/2 & 1 & 0 \\ 0 & -3 & -2 & -1 & 0 & 1 \end{bmatrix}$$

Elemen A(2,2) sekarang menjadi Pivot, untuk menolkan A(3,2). Untuk itu baris kedua kita kalikan dengan -1, untuk kemudian baris ketiga ditambah tiga kali baris kedua tersebut.

$$\begin{bmatrix} 1 & 2 & 2 & 1/2 & 0 & 0 \\ 0 & 1 & 3 & 3/2 & -1 & 0 \\ 0 & 0 & 7 & 7/2 & -3 & 1 \end{bmatrix}$$

Elemen A(3,3) menjadi Pivot. Dan seterusnya diperoleh :

$$\begin{bmatrix} 1 & 2 & 2 & 1/2 & 0 & 0 \\ 0 & 1 & 3 & 3/2 & -1 & 0 \\ 0 & 0 & 1 & 1/2 & -3/7 & 1/7 \end{bmatrix}$$

Elemen A(3,3) ini kemudian menolkan A(2,3) dan A(1,3).

$$\begin{bmatrix} 1 & 2 & 0 & -1/2 & 6/7 & -2/7 \\ 0 & 1 & 0 & 0 & 2/7 & -3/7 \\ 0 & 0 & 1 & 1/2 & -3/7 & 1/7 \end{bmatrix}$$

Terakhir A(2,2) menolkan A(1,2)

$$\begin{bmatrix} 1 & 0 & 0 & -1/2 & 2/7 & 4/7 \\ 0 & 1 & 0 & 0 & 2/7 & -3/7 \end{bmatrix}, \text{Matriks A sudah berubah} \begin{bmatrix} 0 & 0 & 1 & 1/2 & -3/7 & 1/7 \end{bmatrix}$$

Matriks A sudah berubah menjadi Matriks Identitas.

Matriks Invers A^{-1} diperoleh adalah

$$\begin{bmatrix} 1/2 & 2/7 & 4/7 \\ 0 & 2/7 & -3/7 \\ 1/2 & -3/7 & 1/7 \end{bmatrix}$$

CATATAN : Di sini kita tidak menggunakan maksimum pivot. Untuk mendapatkan ketelitian yang tinggi, maksimum pivot dapat kita gunakan, kesulitan yang timbul adalah waktu pertukaran kolom. Kolom elemen A(M,M) yang akan diambil sebagai pivot bernilai = 0, kita lakukan pertukaran baris sehingga pivot tersebut <> 0.

Program 3.8

```
  5   REM PROGRAM
 10   REM MATRIKS INVERS DENGAN ELIMINASI GAUSS-JOORDAN
 15   INPUT "UKURAN MATRIKS"; N
 20   DIM B(N,2*N), A(N,N)
 30   FOR Z=1 TO N : FOR J=1 TO N : INPUT A(I,J) : NEXT J,I
 40   FOR I = 1 TO N
 50   FOR J = 1 TO 2 * N
 60     IF J < N + 1 THEN G(I,J) = A(I,J) : GO TO 90
 70     IF J = I + N THEN G(I,J) = 1 : GO TO 90
 80     G(I,J) = 0
 90   NEXT J
100    NEXT I
110    FOR I = 1 TO N
120    FOR J = I TO 2 * N
130      G(I,J) = G(I,J)/ G(I,I)
140    NEXT J
145    IF I = N THEN 220
150    FOR K = I + 1 TO N
160      R = G(K,I)
170      FOR J = I TO 2 * N
180      G (K,J) = G(K,J) - R * G (K,J) (I,J)
190      NEXT J
200   NEXT K
210   NEXT I
220   FOR I = N TO 2 STEP -1
230     FOR K = I -1 TO 1 STEP -1
240     R = G(K,I)
250       FOR J = I TO 2 * N
260       G(K,J) = G(K,J) - R * G(I,J)
270       NEXT J
280     NEXT K
290   NEXT I
300  REM MENCETAK INVERS
```

```
310    FOR I = 1 TO N
320       FOR J = N+1 TO 2 * N
330       PRINT TAB (2*(J-N)); G(I,J);
340       NEXT J
350    PRINT
360    NEXT I
400    END
```

LATIHAN SOAL : Lakukan modifikasi pada program di atas bila kita gunakan maksimum pivot.

3.7 PERSAMAAN LINIER SIMULTAN DAN ELIMINASI GAUSS

Untuk menyelesaikan persamaan linier simultan kita dapat menggunakan Metode Eliminasi Gauss, seperti halnya mencari Matriks Invers di atas.

Sebagai contoh, kita selesaikan persamaan linier simultan berikut ini :

$$\begin{bmatrix} 2x_1 + 4x_2 + 4x_3 = 13 \\ 3x_1 + 5x_2 + 3x_3 = 12.5 \\ 2x_1 + x_2 + 2x_3 = 6 \end{bmatrix}$$

Persamaan di atas dapat ditulis

$$\begin{bmatrix} 2 & 4 & 4 & x_1 & 13 \\ 3 & 5 & 3 & x_2 & 12.5 \\ 2 & 1 & 2 & x_3 & 6 \end{bmatrix}$$

Selanjutnya kita tulis Matriks sebagai berikut :

$$\begin{bmatrix} 2 & 4 & 4 & 13 \\ 3 & 5 & 3 & 12.5 \\ 2 & 1 & 2 & 6 \end{bmatrix}$$

Tujuan kita menjadikan Submatriks (3x3), sebelah kiri, menjadi Matriks identitas, sama seperti ketika mencari Invers.

Perhatikan hasil-hasil berikut :

$$\begin{bmatrix} 2 & 4 & 4 & 13 \\ 3 & 5 & 3 & 12.5 \\ 2 & 1 & 2 & 6 \end{bmatrix} \begin{bmatrix} 1 & 2 & 2 & 6.5 \\ 3 & 5 & 3 & 12.5 \\ 2 & 1 & 2 & 6 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 2 & 6.5 \\ 0 & -1 & -3 & -7 \\ 0 & -3 & -2 & -7 \end{bmatrix} \begin{bmatrix} 1 & 2 & 2 & 6.5 \\ 0 & 1 & 3 & 7 \\ 0 & -3 & -2 & -7 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 2 & 6.5 \\ 0 & 1 & 3 & 7 \\ 0 & 0 & 7 & 14 \end{bmatrix} \begin{bmatrix} 1 & 2 & 2 & 6.5 \\ 0 & 1 & 3 & 7 \\ 0 & 0 & 1 & 2 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 0 & 2.5 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & 2 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & 0.5 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & 2 \end{bmatrix}$$

Diperoleh $x^1 = 0.5$, $x^2 = 2$

Dengan melihat kemiripan mencari Matriks Invers dengan solusi persamaan linier, silakan anda buat program untuk menyelesaikan persamaan linier tersebut. Lebih unik lagi apabila digunakan maksimum pivot.

ITERASI GAUSS-SEIDEL

Untuk menyelesaikan persamaan linier Simultan dengan iterasi Gauss-Seidel, perhatikan contoh persamaan linier.

$$x_1 + x_2 = 2$$
$$3x_1 - 10x_2 = 3$$

Langkah Penyelesaian :

Ambil dugaan awal untuk x_2, misalnya $x_2 = 0$>

Dari persamaan (1) diperoleh $x_1 = 2$. Substitusi ke persamaan (2) menghasilkan $x_2 = 0.3$. Harga $x_1 = 2$ dan $x_2 = 0.3$ adalah hasil iterasi pertama.

Selanjutnya $x_2 = 0.3$ disubstitusikan lagi ke persamaan (1), diperoleh $x_1 = 1.7$ yang bila disubstitusikan ke persamaan (2) menghasilkan $x_2 = .21$. Harga $x_1 = 1.7$ dan $x_2 = .21$ adalah hasil iterasi kedua.

Demikian dilakukan iterasi selanjutnya sampai selisih antara dua iterasi yang berurutan memenuhi kriteria kita.

Program 3.8

```
60   FOR I = 1 TO N
70   W(I) = 0
80   NEXT I
85   REM ITERASI PERTAMA
90   K = 1
100  FOR I = 1 TO N
110  S = 0
120  FOR J = 1 TO N
130     IF I = J THEN 150
140     S = A(I,J) * W(J)
150  NEXT J
160  X(I,K) = B(I) - S
170  NEXT I
```

```
180   K = K + 1 : REM ITERASI BERIKUTNYA
190   FOR I = 1 TO N
200   S = 0
210   FOR J = 1 TO N
220      IF I = J THEN 240
230      S = A(I,J) * X(J, K-1)
240   NEXT J
250   X(I,K) = B((I) - S)
260   NEXT I
265   REM UJI KONVERGEN
270   FOR J = 1 TO N
280   IF ABS(X(J,K) - X(J,K-1)) > .001 THEN 300 ELSE 350
290   NEXT J
300   REM SUDAH KONVERGEN
310   FOR J = 1 TO N
320   PRINT X(J,K)
330   NEXT J
340   STOP
350   REM BELUM KONVERGEN
360   IF K < M THEN 180
370   PRINT "SUDAH ";M;" ITERASI"
380   GOTO 310
400   END
```

LATIHAN SOAL : Sempurnakan program di atas sehingga kita dapat mengetahui apabila iterasi ini divergen.

3.8. EIGENVALUE

Diketahui Matriks bujursangkar A berukuran (N x N). Bilangan k disebut *Eigenvalue* dari A, bila ada vektor(Nx1) v <>0, sehingga AV = kV. V disebut *Eigenvektor* yang bersangkutan dengan Eigenvalue k.
Bila Matriks A Simetris, maka Eigenvaluenya selalu real.

Untuk Matriks (2 x 2), misalnya $A = \begin{matrix} 4 & 2 \\ 2 & 1 \end{matrix}$ kita dapat menghitung eigenvalue melalui *Persamaan Karakteristik*

$$\begin{vmatrix} 4-k & 2 \\ 2 & 1-k \end{vmatrix} = 0 \quad \text{atau} \quad k^2 - 5*k = 0,$$

diperoleh

$$k_1 = 0, k_2 = 5.$$

Perhatikan bahwa bila S = A(1,1) + A(2,2)

dan $D = \det(A) = \begin{vmatrix} 4 & 2 \\ 2 & 1 \end{vmatrix} = 0$

maka persamaan karakteristik secara umum berbentuk :

$k^2 - S*k + D = 0$

*LATIHAN SOAL:*Buatlah program untuk menghitung Eigenvalue Matriks Bujursangkar berukuran (2 x2)

Secara umum kita dapat mencari koefisien persamaan karakteristik dengan *Metode Bocher.*

Perhatikan contoh berikut :

Diketahui $A = \begin{bmatrix} 1 & 2 & 1 \\ 2 & 3 & 2 \\ 1 & 2 & 4 \end{bmatrix}$

Langkah Penyelesaian :

Sebut $A_1 = A$

Hitung T_1 = Trace (A_1) = jumlah elemen diagonal utama

$= A_1(1,1) + A_1(2,2) + A_1(3,3)$

$= 1 + 3 + 4 = 8$

Tentukan Matriks B_1yang diperoleh dari A_1dengan mengurangi diagonal utama A_1denganT_1

$$B_1 = \begin{bmatrix} -7 & 2 & 1 \\ 2 & -5 & 2 \\ 1 & 2 & -4 \end{bmatrix}$$

Hitung Matriks $A_2 = A_2*B_1 =$

$$\begin{bmatrix} 2 & -6 & 1 \\ 6 & -7 & 0 \\ 1 & 0 & -11 \end{bmatrix}$$

Hitung T_2 = 1/2 * Trace (A_2) = 1/2 * (-2-7-11) = -10

Matriks B_2 diperoleh dari A_2 dengan mengurangi diagonal utama A_1 dengan T_1

$$B_2 = \begin{bmatrix} 8 & -6 & -1 \\ 6 & 3 & 0 \\ 1 & 0 & -1 \end{bmatrix}$$

Hitung Matriks $A_3 = AB_2 =$

$$\begin{bmatrix} 3 & 0 & 0 \\ 0 & -3 & 0 \\ 0 & 0 & -3 \end{bmatrix}$$

T_3 = 1/3 * Trace (A_3) = 1/3 * (-3 * 3 -3) = -3

Persamaan karakteristik memenuhi rumus :

$$k^3 - T_1*k^2 - T_2*k - T_3 = 0$$

atau: $$k^3 - 8*k^2 + 10*k + 3 = 0$$

Dengan mencari akar persamaan di atas, diperoleh eigenvalue yang ditanyakan.

Berikut ini subprogramnya :

Program 3.9

```
10  REM METODE BOCMER/ /FADEV-LEVERRIER
15  REM UNTUK KOEFISIEN PERSAMAAN KARAKTERISTIK
20  REM MENGHITUNG TRACE DAN T(K)
80  FOR I = 1 TO N
90  FOR J = 1 TO N
95  AK(I,J) = A(I,J)
100 NEXT J
105 NEXT I
115 TRACE = 0
120 FOR I = 1 TO N
130 TRACE = TRACE + AK(I,I)
140 NEXT I
150 T(K) = TRACE/K
152 IF K = N THEN 250
155 REM MENENTUKAN MATRIKS BK
160 FOR I = 1 TO N
170 FOR J = 1 TO N
180  IF I = J THEN BK(I,J) = AK(I,J) - T(K) : GO TO 200
190  BK(I,J) = AK(I,J)
200  NEXT J
210  NEXT I
220  GOSUB 500 : REM PERKALIAN MATRIKS A * BK UNTUK MENDAPATKAN
    AK YANG BARU
230  K = K + 1
240  GO TO 115
250  PRINT "KOEFISIEN PERSAMAAN KARAKTERISTIK :"
260  PRINT
270  FOR M = 1 TO N
280  PRINT -T(K)
290  NEXT K
300 STOP
```

Akhirnya cara Bocher tersebut dapat kita kombinasikan dengan Metode Newton Raphson untuk mencari semua Eigenvalue, dan kita gunakan Eliminasi Gauss untuk menghitung Eigenvektornya.

Berikut ini Programnya.

Program 3.10

```
1    SCREEN 0:WIDTH 80:CLS:KEY OFF
5    PRINT"*********************************************************     "
10   PRINT"*                                                               *"
15   PRINT"*      CONTOH PROGRAM                                           *"
20   PRINT"*                                                               *"
25   PRINT"*      PERHITUNGAN EIGENVALUE DAN EIGENVECTOR
     MENGGUNAKAN                                                       *"
30   PRINT"*                                                               *"
35   PRINT"*      METODE BOCHER(FADDEEV-LEVERRIER)/NEWTON-RA
     PHSON/GAUSS                                                       *"
40   PRINT"*                                                               *"
90   PRINT"*********************************************************"
100  '************************ Program Utama************************ "
110  '
120  'Enter order matriks dan elemennya
130  '
140  PRINT: INPUT"ENTER ORDER MATRIKS";N
150  DIM  A(N,N+1),B(N,N),C(N,N),X(N),ALPHA(N),AA(N),BB(N),XW(N),
     NPIVROW(N-1,2),NPIVCOL(N-1,2)
160  CLS:PRINT:PRINT "ENTER ELEMEN MATRIKS:"
170  FOR K = 1 TO N
180  PRINT : PRINT "BARIS ";K :PRINT
190  FOR J = 1 TO N
200  PRINT "   ELEMEN (";K;",";J;") =";: INPUT B(K,J)
210  NEXT J
220  NEXT K
230  NN=N: NC=N
240  TRC=0
250  FOR K = 1 TO N
260  FOR J = 1 TO N
270  A(K,J) = B(K,J)
280  NEXT J
290  TRC=TRC + B(K,K)
300  NEXT K
310  PRINT:PRINT
320  PRINT"*********************************************************     "
330  PRINT" MATRIX SEMULA:"
340  GOSUB 2000 'Print matrix
350  PRINT: INPUT "APAKAH MATRIX BENAR(Y/T)"; Q$:PRINT
360  IF Q$ = "Y" OR Q$ = "y" THEN 410
370  PRINT"POSISI ELEMEN YANG DIKOREKSI:":PRINT
380  INPUT " BARIS NOMOR";NROW :INPUT "   KOLOM NOMOR";NCOL
390  PRINT:INPUT "   NILAI YANG BENAR"; B(NROW,NCOL):PRINT
```

```
400   GOTO 240
410   'Memulai prosedur Faddeev-Leverrier
420   PRINT CHR$(12)
430   PRINT"************METODEFADDEEV-LEVERRIER*************** "
440   PRINT
450   PRINT "INGIN MENGIKUTI STEP-BY-STEP(Y/T) ";
460   INPUT Q2$:PRINT
470   PRINT"*********************************************** "
480   FOR K = 2 TO N + 1
490   TRACE=0
500   FOR I = 1 TO N
510   TRACE=TRACE + A(I,I)
520   NEXT I
530   ALPHA(K-1)=TRACE/(K-1)
540   IF K=N+1 GOTO 750
550   FOR I = 1 TO N
560   FOR J = 1 TO N
570   IF I=J THEN FACT=1 ELSE FACT=0
580   A(I,J)=A(I,J) - FACT*ALPHA(K-1)
590   NEXT J: NEXT I
600   IF Q2$="Y" OR Q2$="y" THEN GOSUB 2000
610   FOR I = 1 TO N
620   FOR J = 1 TO N
630   SUM=0
640   FOR L = 1 TO N
650   SUM = SUM + B(I,L)*A(L,J)
660   NEXT L
670   C(I,J)=SUM
680   NEXT J:NEXT I
690   FOR I = 1 TO N
700   FOR J = 1 TO N
710   A(I,J)=C(I,J)
720   NEXT J:NEXT I
730   IF Q2$="Y" OR Q2$="y" THEN GOSUB 2000
740   NEXT K
750   PRINT"*********************************************** "
760   PRINT:PRINT "KOEFISIEN POLINOM KARAKTERISTIK:": PRINT
770   FOR J = 1 TO N
780   PRINT "ALPHA(";J;") =";ALPHA(J)
790   NEXT J
800   PRINT
810   PRINT"*********************************************** "
820   PRINT"*********************************************** "
830   PRINT:INPUT"TEKAN ENTER UNTUK MELANJUTKAN",KK$
```

```
840  PRINT CHR$(12)
850  PRINT"**************METODENEWTON-RAPHSON*************** "
860  AA(N)=1
870  FOR I=1 TO N
880  AA(N-I)=-ALPHA(I)
890  NEXT I
900  NW=N
910  PRINT:PRINT"BERIKAN DUGAAN TERBAIK ANDA TENTANG
     EIGENVALUE";:INPUT X1
920  PRINT:PRINT"BERIKAN NILAI KONVERGENSI DARI POLINOM";: INPUT
     EPS
930  GOSUB 3000 'Rutin untuk mencari Akar
940  XW(N) = X1
950  '
960  'Pembagian Sintetik
970  IF N = 1 GOTO 1260
980  BB(N-1)=AA(N)
990  FOR R = 1 TO N - 1
1000 BB(N - 1 - R) = AA(N - R) + BB(N - R) * X1
1010 NEXT
1020 N = N - 1
1030 FOR JJ = N TO 0 STEP  - 1
1040 AA(JJ) = BB(JJ)
1050 NEXT
1060 '
1070 'Memeriksa Akar Kmpleks
1080 IF N<>2 GOTO 1220
1090 WW=AA(1)^2-4*AA(2)*AA(0)
1100 IF WW>=0 GOTO 1220
1110 XW(2)=-AA(1)/(2*AA(2))
1120 XW(1)=SQR(-WW)/(2*AA(2))
1130 PRINT "KE ";NW;" EIGENVALUE ADALAH:":PRINT
1140 FOR JK = NW TO 3 STEP -1
1150 PRINT XW(JK)
1160 NEXT
1170 '
1180 'Print Akar Kompleks
1190 PRINT XW(2);"+";XW(1);"i"
1200 PRINT XW(2);"-";XW(1);"i"
1210 PRINT"EIGENVALUE ADALAH KOMPLEKS":GOTO 1890
1220 FOR PAUSE=1 TO 1000:NEXT PAUSE
1230 IF N > 0 GOTO 930
1240 '
1250 'Print Akar
```

```
1260  PRINT:PRINT"KE ";NW;" EIGENVALUE ADALAH:":PRINT
1270  EIGENSUM=0
1280  FOR I=NW TO 1 STEP -1
1290  PRINT"  ";XW(I)
1300  EIGENSUM=EIGENSUM + XW(I)
1310  NEXT I
1320  PRINT:PRINT"PENYELIDIKAN:"
1330  PRINT:PRINT"PENJUMLAHAN EIGENVALUE = ";EIGENSUM
1340  PRINT"TERAS (TRACE) DARI MATRIX  = ";TRC
.1350 PRINT"SELISIH ADALAH           = ";EIGENSUM-TRC : PRINT
1360  IF ABS(EIGENSUM-TRC)>EPS THEN PRINT"SELISIH LEBIH BESAR DARI
      KRITERIA KONVERGENS{" : PRINT
1370  PRINT"*********************************************************** "
1380  N=NW
1390  PRINT:PRINT:INPUT"TEKAN ENTER UNTUK MELANJUTKAN",KK$
1560  PRINT CHR$(12)
1570  PRINT"*************METODE ELIMINASI GAUSS**************** "
1580  PRINT
1590  PRINT"BERIKAN NILAI MINIMUM ELEMEN PIVOT YANG DIPERBO
      LEHKAN UNTUK                                                    "
1600  PRINT"PROSEDUR ELIMINASI GAUSS ";:INPUT EPS:PRINT
1610  PRINT"INGIN MELIHAT STEP-BY-STEP";
1620  PRINT"PROSEDUR ELIMINASI GAUSS(Y/T)";:INPUT Q2$:PRINT
1630  FOR KK = N TO 1 STEP -1
1640  PRINT"*********************************************************** "
1650  PRINT"HIMPUNAN HOMOGEN (A - EIGENVALUE*I)*X = 0, UNTUK
      EIGENVALUE =";XW(KK)
1660  NC=N+1
1670  'Kurangkan eigenvalue dari matrix dan gunakan metode Gauss
1680  FOR I = 1 TO N
1690  FOR J = 1 TO N
1700  A(I,J) = B(I,J)
1710  NEXT J
1720  A(I,I)= B(I,I) - XW(KK)
1730  A(I,NC)=0
1740  NEXT I
1750  GOSUB 2000 'Print matrix
1760  GOSUB 9000 'Pelaksanaan Eliminasi
1770  NEXT KK
1780  PRINT:PRINT
1790  PRINT"*********************************************************** "
1800  PRINT:PRINT'ANDAINGINMENGULANGIPERHITUNGAN": PRINT
      "DENGANSEDIKIT PERUBAHAN KOEFISIEN(Y/T)";:INPUT V$
1810  IF V$ = "Y" OR V$ = "y" THEN 1820 ELSE 1830
```

```
1820  CLS : GOTO 230
1830  PRINT:INPUT "ANDA INGIN MEMPROSES LAGI(Y/T)";W$
1840  IF W$ = "Y" OR W$ = "y" THEN 1850 ELSE 1880
1850  PRINT:INPUT "APAKAH ORDER TETAP SAMA";WW$
1860  IF WW$="T" OR WW$="t" THEN PRINT CHR$(12):RUN 100
1870  CLS : GOTO 160
1880  PRINT:PRINT
1890  PRINT"****************** PROGRAM SELESAI ****************** "
1900  END
2000  '*************** Subrouline 1: Print matrix ************************ "
2010  '
2020  FOR KA = 1 TO N
2030  PRINT
2040  FOR J = 1 TO NC
2050  PRINT A(KA,J),
2060  NEXT J:PRINT:NEXT KA:PRINT
2070  PRINT"**************************************************** "
2090  RETURN
3000  '*************** Subroutine 2: Metode Newton-Raphson *****************
3010  '
3020  PRINT:PRINT"KONVERGEN KE EIGENVALUE";(NW-N+1):PRINT
3030  FOR ITER=0 TO 40 STEP 2
3040  X=X1:GOSUB 4000
3050  PRINT TAB(3) "x=";X1 TAB(20) "f(x)=";F
3080  IF  ABS (F)< EPS GOTO 3120
3090  X1=X1-F/FP
3100  NEXT ITER
3110  PRINT"MELAMPAUI LIMIT ITERASI:TIDAK KONVERGEN":
PRINT"ITER=";ITER:PRINT"EPS=";EPS:PRINT"F=";F:  PRINT"FUNCTION
KEY F5 AKAN MELANJUTKAN PROGRAM":STOP
3120  RETURN
4000  '***** Subroutine 3: Evaluasi polinom dan derivatifnya *****
4010  '
4020  F = 0:FP=0
4030  FOR KK = N TO 0 STEP  - 1
4040  F = F + AA(KK) * X ^ KK
4050  IF KK=0 GOTO 4080
4060  FP= FP+ KK*AA(KK) * X ^ (KK-1)
4070  NEXT KK
4080  RETURN
9000  '*************** Subroutine 4: Metode Eliminasi Gauss *************** "
9010  '
9020  FOR K = 1 TO N-1
9030  'Menggunakan stategi pivoting lengkap
```

```
9040  MAXPIVOT = ABS(A(K,K))
9050  NPIVROW(K,1)=K: NPIVROW(K,2)=K
9060  NPIVCOL(K,1)=K: NPIVCOL(K,2)=K
9070  FOR I = K TO N
9080  FOR J = K TO N
9090  IF MAXPIVOT > = ABS(A(I,J)) GOTO 9130
9100  MAXPIVOT=ABS(A(I,J))
9110  NPIVROW(K,1)=K: NPIVROW(K,2)=I
9120  NPIVCOL(K,1)=K: NPIVCOL(K,2)=J
9130  NEXT J:NEXT I
9140  IF MAXPIVOT > = EPS GOTO 9160
9150  GOTO 9420
9160  IF NPIVROW(K,2)=K GOTO 9230
9170  IF Q2$="Y" OR Q2$="y" THEN PRINT"PIVOT BARIS:"
9180  IFQ2$="Y"ORQ2$="y"THENPRINT "PERTUKARANBARIS
      ";NPIVROW(K,2);" DAN ";K
9190  FOR J = K TO NC
9200  SWAP  A(NPIVROW(K,2),J),A(K,J)
9210  NEXT J
9220  IF Q2$="Y" OR Q2$="y" THEN GOSUB 2000 'Print matrix
9230  IF NPIVCOL(K,2)=K GOTO 9300
9240  IF Q2$="Y" OR Q2$="y" THEN PRINT"PIVOT KOLOM:"
9250  IF Q2$="Y" OR Q2$="y" THEN PRINT "PERTUKARAN
      KOLOM";NPIVCOL(K,2);" AND ";K
9260  FOR I = 1 TO N
9270  SWAP  A(I,NPIVCOL(K,2)),A(I,K)
9280  NEXT I
9290  IF Q2$="Y" OR Q2$="y" THEN GOSUB 2000 'Print matrix
9300  IF Q2$="Y" OR Q2$="y" THEN PRINT "KERJAKAN ELIMINASI:"
9310  FOR I = K+1 TO N
9320  IF Q2$="Y" OR Q2$="y" THEN PRINT "BAGI BARIS ";K;" DENGAN ";A(K,K)
9330  IF Q2$="Y" OR Q2$="y" THEN PRINT "KALIKAN BARIS ";K;" DENGAN
      ";A(I,K) ;"DAN KURANGKAN DARI BARIS ";I
9340  MULT = - A(I,K)/A(K,K)
9350  FOR J = NC TO K STEP -1
9360  A(I,J) = A(I,J) + MULT * A(K,J)
9370  NEXT J
9380  IF Q2$="Y" OR Q2$="y" THEN GOSUB 2000 'Print matrix
9390  NEXT I: NEXT K
9400  '
9410  'APemakaian formula substitusi kembali
9420  RANK=K-1 :PRINT"RANK =";RANK:NMR=N-RANK
9430  PRINT"PROGRAM MENJADIKAN ";NMR;" ELEMENT EIGENVEKTOR
      MENJADI SATUAN(UNITY)"
```

```
9440 PRINT"DAN MEREDUKSIKAN PROBLEMA MENJADI PENENTUAN
     ";RANK;"ELEMENT."
9450 FOR JJ=1 TO NMR : X(N+1-JJ) = 1: NEXT JJ
9460 FOR I = RANK TO 1 STEP -1
9470 SUM = 0
9480 FOR J = I+1 TO N
9490 SUM = SUM + A(I,J) * X(J)
9500 NEXT J
9510 X(I) = (A(I,NC) - SUM) / A(I,I)
9520 IF ABS(X(I)) < EPS THEN X(I)=0
9530 NEXT I
9540 '
9550 'Menukar urutan anu
9560 FOR K=N-1 TO 1 STEP -1
9570 SWAP X(NPIVCOL(K,2)), X(NPIVCOL(K,1))
9580 NEXT K
9590 '
9600 PRINT"********************************************************* "
9610 PRINT
9620 PRINT "EIGENVEKTOR bersangkutan dengan EIGENVALUE =";XW(KK):
     PRINT
9630 FOR J = 1 TO N
9640 PRINT "X(";J;") =";X(J)
9650 NEXT J
9660 PRINT
9670 PRINT"********************************************************* "
9680 PRINT:PRINT"PENYELIDIKAN:"
9690 PRINT:PRINT"A*X          =  EIGENVALUE*X":PRINT
9700 FOR I=1 TO N
9710 AX=0
9720 FOR J=1 TO N
9730 AX=AX + B(I,J)*X(J)
9740 NEXT J
9750 EX=XW(KK)*X(I)
9760 PRINT AX TAB(20) EX
9770 NEXT I
9780 PRINT
9790 PRINT"********************************************************* "
9800 PRINT:INPUT"TEKAN ENTER UNTUK MELANJUTKAN ";KK$
9810 PRINT CHR$(12)
9820 RETURN
```

Sebagai penutup dari Bab 3 ini, disajikan berturut-turut 3 buah Program yang cukup lengkap mengenai Eliminasi Gauss, Eliminasi Gauss-Jordan, dan Iterasi Gauss-Seidel.

Program 3.11

```
1   SCREEN 0:WIDTH 80:CLS:KEY OFF
5   PRINT "********************************************************* "
10  PRINT"*                                                         * "
15  PRINT"*      CONTOH PROGRAM                                     * "
20  PRINT"*                                                         * "
25  PRINT"*      METODE ELIMINASI GAUSS                             * "
30  PRINT"*                                                         * "
35  PRINT"*      UNTUK PERSAMAAN LINIER SIMULTAN                    * "
50  PRINT"*                                                         * "
90  PRINT"********************************************************* "
100 '*********************** Program Utama ************************
110 .'
120 'Enter banyak persamaan, koefisien, dan konstanta
130 '
140 PRINT:PRINT "BANYAK PERSAMAAN";:INPUT N
150 DIM A(N,N+1), B(N,N+1), X(N), NPIVROW(N,2),NPIVCOL(N,2)
160 PRINT:PRINT "ENTER KOEFISIEN DAN KONSTANTA SETIAP PERSA-
    MAAN"
170 FOR K = 1 TO N
180 PRINT : PRINT "PERSAMAAN ";K: PRINT
190 FOR J = 1 TO N
200 PRINT "   KOEFISIEN (";K;",";J;") =";: INPUT B(K,J)
210 NEXT J
220 PRINT:PRINT "   KONSTANTA ";K;" =";: INPUT B(K,N+1)
230 NEXT K
240 NC=N+1
250 PRINT
260 PRINT"GBERIKANNILAIMINIMUMPIVOTYANG
    DIPERKENANKAN";:INPUT      EPS
270 PRINT CHR$(12)
280 DET=1
290 FOR K = 1 TO N
300 FOR J = 1 TO NC
310 A(K,J)=B(K,J)
320 NEXT J : NEXT K
330 PRINT:PRINT
340 PRINT"********************************************************* "
350 PRINT"MATRIX LENGKAP:"
360 GOSUB 2000
370 PRINT: INPUT "APAKAH MATRIX SUDAH BENAR(Y/T)"; Q$:PRINT
380 IF Q$ = "Y" OR Q$ = "y" THEN 450
390 PRINT"BERIKAN POSISI ELEMEN YANG DIPERBAIKI:":PRINT
400 INPUT "   NOMOR BARIS";NROW :INPUT "   NOMOR KOLOM";NCOL
```

```
410  PRINT:INPUT "   NILAI YANG BENAR"; B(NROW,NCOL):PRINT
420  GOTO 270
430  '
440  'Mulai dengan Prosedur Eliminasi Gauss
450  INPUT "INGIN MELIHAT HASIL STEP-BY-STEP (Y/T)";Q2$:PRINT
460  PRINT"*******************************************************   "
470  FOR K = 1 TO N
480  'GUNAKAN strategi pivoting lengkap
490  MAXPIVOT = ABS(A(K,K))
500  NPIVROW(K,1)=K: NPIVROW(K,2)=K
510  NPIVCOL(K,1)=K: NPIVCOL(K,2)=K
520  FOR I = K TO N
530  FOR J = K TO N
540  IF MAXPIVOT > = ABS(A(I,J)) GOTO 580
550  MAXPIVOT=ABS(A(I,J))
560  NPIVROW(K,1)=K: NPIVROW(K,2)=I
570  NPIVCOL(K,1)=K: NPIVCOL(K,2)=J
580  NEXT J:NEXT I
590  IF MAXPIVOT > = EPS GOTO 610
600  PRINT"PIVOT LEBIH KECIL DARI";EPS;". MATRIX BOLEH JADI
     SINGULAR.":GOTO 910
610  IF NPIVROW(K,2)=K GOTO 690
620  IF Q2$="Y" OR Q2$="y" THEN PRINT"PIVOT BARIS:"
630  IF Q2$="Y" OR Q2$="y" THEN PRINT "PERTUKARAN BARIS
     ";NPIVROW(K,2);"    DAN ";K
640  FOR J = K TO NC
650  SWAP  A(NPIVROW(K,2),J),A(K,J)
660  NEXT J
670  DET=DET*(-1)
680  IF Q2$="Y" OR Q2$="y" THEN GOSUB 2000
690  IF NPIVCOL(K,2)=K GOTO 770
700  IF Q2$="Y" OR Q2$="y" THEN PRINT"PIVOT KOLOM:"
710  IF Q2$="Y" OR Q2$="y" THEN PRINT "PERTUKARAN KOLOM
     ";NPIVCOL(K,2);" DAN ";K
720  FOR I = 1 TO N
730  SWAP  A(I,NPIVCOL(K,2)),A(I,K)
740  NEXT I
750  DET=DET*(-1)
760  IF Q2$="Y" OR Q2$="y" THEN GOSUB 2000
770  IF K=N THEN GOTO 880
780  IF Q2$="Y" OR Q2$="y" THEN PRINT "LAKUKAN ELIMINASI:"
790  FOR I = K+1 TO N
800  IF Q2$="Y" OR Q2$="y" THEN PRINT "BAGI BARIS ";K;" DENGAN ";A(K,K)
810  IF Q2$="Y" OR Q2$="y" THEN PRINT "KALIKAN BARIS ";K;" DENGAN
```

```
     ";A(I,K) ;"DAN KURANGKAN DARI BARIS ";I
820  MULT = - A(I,K)/A(K,K)
830  FOR J = NC TO K STEP -1
840  A(I,J) = A(I,J) + MULT * A(K,J)
850  NEXT J
860  IF Q2$="Y" OR Q2$="y" THEN GOSUB 2000
870  NEXT I
880  NEXT K
890  '
900  'Gunakan formula substitusi kembali
910  RANK=K-1 :PRINT"RANK =";RANK:NMR=N-RANK
920  IF RANK=N THEN X(N) = A(N,N+1) / A(N,N) :NCOUNT=N-1: GOTO 970
930  PRINT"PROGRAM MENJADIKAN ";NMR;" ANU MENJADI =1,"
940  PRINT"DAN REDUKSI PROBLEMA PENENTUAN ";RANK;" ANU."
950  FOR JJ=1 TO NMR : X(N+1-JJ) = 1: NEXT JJ
960  NCOUNT=RANK
970  FOR I = NCOUNT TO 1 STEP -1
980  SUM = 0
990  FOR J = I+1 TO N
1000 SUM = SUM + A(I,J) * X(J)
1010 NEXT J
1020 X(I) = (A(I,N+1) - SUM) / A(I,I)
1030 NEXT I
1040 '
1050 'Mengubah urutan Anu
1060 FOR K=N TO 1 STEP -1
1070 SWAP X(NPIVCOL(K,2)), X(NPIVCOL(K,1))
1080 NEXT K
1090 '
1100 'Menghitung determinan matrix
1110 FOR I=1 TO N
1120 DET=DET*A(I,I)
1130 NEXT I
1140 PRINT"**********************************************************  "
1150 PRINT:PRINT "HASIL SUBSTITUSI KEMBALI": PRINT
1160 FOR J = 1 TO N
1170 PRINT "X(";J;") =";X(J)
1180 NEXT J
1190 PRINT:PRINT"NILAI DETERMINAN=";DET:PRINT
1200 PRINT"**********************************************************  "
1210 PRINT"**********************************************************  "
1220 PRINT:PRINT "INGIN MENGULANG PERHITUNGAN ":PRINT"DENGAN
     SEDIKIT PERUBAHAN  KOEFISIEN(Y/T)";:INPUT V$
1230 IF V$ = "Y" OR V$ = "y" THEN 1240 ELSE 1250
```

```
1240  CLS : GOTO 270
1250  PRINT:INPUT "INGIN MEMPROSES LAGI(Y/T)";W$
1260  IF W$ = "Y" OR W$ = "y" THEN 1270 ELSE 1300
1270  PRINT:INPUT "APAKAH ORDER BERUBAH";WW$
1280  IF WW$="T" OR WW$="t" THEN PRINT CHR$(12):RUN 100
1290  CLS : GOTO 160
1300  PRINT:PRINT
1310  PRINT"*****************PROGRAMSELESAI ********************* "
1320  END
2000  '*************** Subroutine 1: Print matrix ***********************
2010  '
2020  FOR KA = 1 TO N
2030  PRINT
2040  FOR J = 1 TO NC
2050  PRINT A(KA,J),
2060  NEXT J:PRINT: NEXT KA:PRINT
2070  PRINT"********************************************************* "
2080  FOR IPAUSE = 1 TO 3000 : NEXT
2090  RETURN
```

Program 3.12

```
10    SCREEN 0:WIDTH 80:CLS:KEY OFF
20    PRINT "******************************************************** "
30    PRINT"*                                                        * "
40    PRINT"*        CONTOH  PROGRAM                                 * "
50    PRINT"*                                                        * "
60    PRINT"*        METODE REDUKSI GAUSS-JORDAN                     * "
70    PRINT"*                                                        * "
80    PRINT"*        UNTUK PERSAMAAN LINEAR SIMULTAN                 * "
90    PRINT"*                                                        * "
100   PRINT"*        DAN MATRIX INVERS                               * "
130   PRINT"*                                                        * "
140   PRINT"********************************************************* "
150   '*********************** Program Utama **********************
160   '
180   PRINT:PRINT"ANDA DAPAT MENGGUNAKAN PRORAM INI UNTUK:"
190   PRINT:PRINT"        1. MENYELESAIKAN PERSAMAAN LINEAR  "
200   PRINT:PRINT"        2. MENCARI INVERS MATRIX           "
210   PRINT:PRINT"        3. MENGERJAKAN KEDUA HAL DI ATAS   "
220   PRINT:INPUT"NOMOR PILIHAN ANDA ***";SEL:CLS
230   '
240   'Enter banyaknya persamaan.koefisien dan konstanta
250   '
260   PRINT: IF SEL =2 THEN INPUT"BANYAK BARIS MATRIX";N
```

```
270  IF SEL <> 2 THEN INPUT"BANYAK PERSAMAAN";N
280  DIM A(N,2*N+1), B(N,N+1), C(N,N),X(N)
290  PRINT:IF SEL = 2 THEN PRINT "ENTER ELEMENT MATRIX" ELSE
     PRINT"ENTER KOEFISIEN DAN KONSTANTA SETIAP PERSAMAAN"
300  FOR K = 1 TO N
310  PRINT : IF SEL = 2 THEN PRINT "BARIS";K ELSE PRINT"PERSAMAAN";K
320  PRINT
330  FOR J = 1 TO N
340  IF SEL = 2 THEN PRINT "   ELEMEN (";K;",";J;") =";: INPUT B(K,J)
350  IF SEL <> 2 THEN PRINT "   KOEFISIEN (";K;",";J;") =";: INPUT B(K,J)
360  NEXT J
370  IF SEL <> 2 THEN PRINT:PRINT "   KONSTANTA ";K;" =";: INPUT B(K,N+1)
380  NEXT K
390  PRINT
400  PRINT"BERIKAN HARGA MINIMUM PIVOT YANG
     DIPERKENANKAN";:INPUT EPS
410  PRINT CHR$(12)
420  FOR K = 1 TO N
430  FOR J = 1 TO N + 1
440  A(K,J) = B(K,J)
450  NEXT J
460  FOR J = N+2 TO 2*N + 1
470  A(K,J) = 0
480  NEXT J
490  A(K,K-1+N+2) = 1
500  NEXT K
510  PRINT:PRINT:PRINT:PRINT
520  PRINT"*********************************************************  "
530  PRINT"MATRIX AUGMENTED :"
540  GOSUB 1170
550  PRINT: INPUT "APAKAH MATRIX AUGMENTED BENAR(Y/T)"; Q$:PRINT
560  IF Q$ = "Y" OR Q$ = "y" THEN 620
570  PRINT"BERIKAN POSISI ELEMEN YANG HARUS 0DIBETULKAN"
580  INPUT "BARIS KE";NROW :INPUT "KOLOM KE";NCOL
590  PRINT:INPUT "   HARGA YANG BENAR"; B(NROW,NCOL):PRINT
600  GOTO 420
610  'Awaal prosedur reduksi Gauss-Jordan
620  INPUT "ANDA INGIN MELIHAT HASIL STEP-BY-STEP (Y/T)";Q2$:PRINT
630  PRINT"*********************************************************  "
640  FOR K = 1 TO N
650  'Gunakan partial pivoting strategy
660  MAXPIVOT = ABS(A(K,K)):NPIVOT=K
670  FOR I = K TO N
680  IF MAXPIVOT > = ABS(A(I,K)) GOTO 700
```

```
690  MAXPIVOT=ABS(A(I,K)) : NPIVOT=I
700  NEXT I
710  IF MAXPIVOT > = EPS GOTO 730
720  PRINT"ELEMEN PIVOT LEBIH KECIL DARI";EPS;". MATRIX BOLEH JADI
     SINGULAR. RANK=";K-1 :GOTO 1150
730  IF NPIVOT = K GOTO 800
740  IF Q2$="Y" OR Q2$="y" THEN PRINT"PARTIAL PIVOTING:"
750  IF Q2$="Y" OR Q2$="y" THEN PRINT "SALING TUKAR BARIS";NPIVOT;"
     DAN ";K
760  FOR J = K TO 2*N + 1
770  SWAP  A(NPIVOT,J),A(K,J)
780  NEXT J
790  IF Q2$="Y" OR Q2$="y" THEN GOSUB 1170
800  IF Q2$="Y" OR Q2$="y" THEN PRINT "LAKUKAN NORMALISASI:"
810  IF Q2$="Y" OR Q2$="y" THEN PRINT "BAGI BARIS ";K;" OLEH ";A(K,K)
820  D=A(K,K)
830  FOR J = 2*N+1 TO K STEP -1
840  A(K,J) = A(K,J)/D
850  NEXT J
860  IF Q2$="Y" OR Q2$="y" THEN GOSUB 1170
870  IF Q2$="Y" OR Q2$="y" THEN PRINT "LAKUKAN REDUKSI:"
880  FOR I = 1 TO N
890  IF I=K GOTO 960
900  MULT=A(I,K)
910  IF Q2$="Y" OR Q2$="y" THEN PRINT "KALIKAN BARIS";K;" DENGAN
     ";A(I,K) ;"DAN KURANGKAN TERHADAP BARIS ";I
920  FOR J = 2*N+1 TO K STEP -1
930  A(I,J) = A(I,J) - MULT* A(K,J)
940  NEXT J
950  IF Q2$="Y" OR Q2$="y" THEN GOSUB 1170
960  NEXT I:NEXT K:IF SEL=2 THEN GOTO 1050
970  PRINT"******************************************************* "
980  PRINT:PRINT "HASILNYA:": PRINT
990  FOR J = 1 TO N
1000 X(J)=A(J,N+1)
1010 PRINT "X(";J;") =";X(J)
1020 NEXT J
1030 PRINT
1040 PRINT"******************************************************* "
1050 IF SEL > 1 THEN GOSUB 1270: GOSUB 1360
1060 PRINT"******************************************************* "
1070 PRINT:PRINT "ANDA INGIN MENGULANGI KALKULASI DENGAN
     SEDIKIT PERUBAHAN PADA KOEFISIEN(Y/T)";:INPUT V$
1080 IF V$ = "Y" OR V$ = "y" THEN 1090 ELSE 1100
```

```
1090  CLS : GOTO 410
1100  PRINT:INPUT "ANDA INGIN MEMBENTUK SISTEM BARU(1), atau
      SELESAI(2)";W$
1110  IF W$ = "1" THEN 1120 ELSE 1150
1120  PRINT:INPUT "APAKAH ORDER SAMA(Y/T)";WW$
1130  IF WW$='T" OR WW$="t" THEN PRINT CHR$(12):RUN 150
1140  CLS : GOTO 290
1150  END
1170  '***************** Subroutine 1: Mencetak matrix ***********************
1180  '
1190  FOR KA = 1 TO N
1200  PRINT
1210  FOR J = 1 TO N + 1
1220  PRINT A(KA,J),
1230  NEXT J:PRINT: NEXT KA:PRINT
1240  PRINT"*******************************************************  "
1250  FOR IPAUSE = 1 TO 3000 : NEXT
1260  RETURN
1270  '********** Subroutine 2: Mencetak invers matrix ***************
1280  'PRINT"INVERS MATRIX:"
1290  FOR KA = 1 TO N
1300  PRINT
1310  FOR J = N+2 TO 2*N + 1
1320  PRINT A(KA,J),
1330  NEXT J:PRINT:NEXT KA:PRINT
1340  PRINT"*******************************************************  "
1350  RETURN
1360  '********* Subroutine 3: Memeriksa produk dari atrix dan inversnya ********
1370  PRINT"PRODUK MATRIX DAN INVERS ADALAHE IDENTITY
      MATRIX:":PRINT
1380  FOR I = 1 TO N
1390  FOR J = 1 TO N
1400  C(I,J)=0
1410  FOR K = 1 TO N
1420  C(I,J)=C(I,J) + B(I,K)*A(K,J+N+1)
1430  NEXT K
1440  PRINT USING "##.####     ";C(I,J),
1450  IF I=J AND ABS(C(I,J)-1)<EPS THEN GOTO 1480
1460  IF I<>J AND ABS(C(I,J))<EPS THEN GOTO 1480
1470  PRINT "PERINGATAN: INVERS MUNGKIN KURANG AKURAT."
1480  NEXT J:PRINT:NEXT I:PRINT
1490  RETURN
```

Program 3.13

```
10   SCREEN 0:WIDTH 80:KEY OFF:CLS
20   PRINT:PRINT:PRINT:PRINT
30   PRINT "****************************************************** "
40   PRINT"*                                                         * "
50   PRINT"*          CONTOH PROGRAM                                 * "
60   PRINT"*                                                         * "
70   PRINT"*          SUBSTITUSI ITERASI GAUSS-SEIDEL                * "
80   PRINT"*                                                         * "
90   PRINT"*         UNTUK SISTEM PERSAMAAN LINEAR DIAGONAL          * "
100  PRINT"*                                                         * "
130  PRINT"****************************************************** "
150  '****************** PROGRAM UTAMA**********************
160  '
170  'Enter banyaknya persamaan, koefisien dan konstanta
180  '
190  PRINT:PRINT "BANYAKNYA PERSAMAAN***";:INPUT N:CLS
200  DIM A(N,N+1), B(N,N+1), X(N),OLDX(N)
210  PRINT:PRINT "ENTER KOEFISIEN DAN KONSTANTA SETIAP PERSA-
     MAAN"
220  FOR K = 1 TO N
230  PRINT : PRINT "PERSAMAAN";K: PRINT
240  FOR J = 1 TO N
250  PRINT "   KOEFISIEN (";K;",";J;") =";: INPUT B(K,J)
260  NEXT J
270  PRINT:PRINT "   KONSTANTA ";K;" =";: INPUT B(K,N+1)
280  NEXT K
290  NC=N+1
300  PRINT
310  PRINT"TENTUKAN KRITERIA KONVERGENSI DARI SOLUSI ";
     :INPUT EPS
320  PRINT CHR$(12)
330  PRINT"TENTUKAN DUGAAN AWAL DARI HARGA A N U :":PRINT
340  FOR I=2 TO N:PRINT "   A N U ";I;:INPUT X(I):NEXT I
350  FOR K = 1 TO N
360  FOR J = 1 TO NC
370  A(K,J) = B(K,J)
380  NEXT J : NEXT K
390  PRINT
400  PRINT"****************************************************** "
410  PRINT"MATRIX AUGMENTED:"
420  GOSUB 790
430  PRINT: INPUT "APAKAH MATRIX AUGMENTED BENAR(Y/T)"; Q$:PRINT
440  IF Q$ = "Y" OR Q$ = "y" THEN 510
```

```
450   PRINT"TULIS POSISI ELEMEN YANG AKAN DIBETULKAN :":PRINT
460   INPUT "   BARIS NOMOR ";NROW :INPUT "   KOLOM NOMOR ";NCOL
470   PRINT:INPUT "   NILAI YANG BENAR "; B(NROW,NCOL):PRINT
480   GOTO 350
490   '
500   'Mulai dengan prosedur substitusi Gauss-Seidel
510   INPUT "ANDA INGIN MELIHAT HASIL STEP-BY-STEP(Y/T)";Q2$:PRINT
520   COUNT=0
530   FOR I=1 TO N
540   OLDX(I)=X(I)
550   SUM=0
560   FOR J=1 TO N
570   IF J=I THEN GOTO 590
580   SUM=SUM + A(I,J)*X(J)
590   NEXT J
600   X(I)=(A(I,NC) - SUM)/A(I,I)
610   DEL=(OLDX(I)-X(I))/X(I)
620   IF ABS(DEL) < EPS THEN COUNT=COUNT + 1
630   NEXT I
640   IF Q2$="Y" OR Q2$="y" THEN GOSUB 900
650   IF COUNT=N GOTO 670
660   GOTO 520
670   GOSUB 900
680   '
690   PRINT:PRINT "APAKAH PERHITUNGAN INGIN DIULANGI DENGAN
      SEDIKIT PERUBAHAN PADA KOEFISIEN(Y/T)";:INPUT V$
700   IF V$ = "Y" OR V$ = "y" THEN 710 ELSE 720
710   CLS : GOTO 350
720   PRINT:INPUT "APAKAH PROSES LAGI DENGAN PERSAMAAN BARU (1)
      , atau SELESAI (2) ";W$
730   IF W$ = "1" THEN 740 ELSE 770
740   PRINT:INPUT "APAKAH SISTEM YANG BARU BERUKURAN SAMA (Y/
      T)";WW$
750   IF WW$="T" OR WW$="t" THEN PRINT CHR$(12):RUN 150
760   CLS : GOTO 210
770   END
790   '**************** Subroutine 1:Mencetak matrix ************************
800   '
810   FOR KA = 1 TO N
820   PRINT
830   FOR J = 1 TO NC
840   PRINT A(KA,J),
850   NEXT J:PRINT
860   NEXT KA:PRINT
```

```
870   PRINT"*******************************************************  "
880   RETURN
890   '
900   '**************** Subroutine 2: Mencetak hasil **********************
910   '
920   PRINT"*******************************************************  "
930   PRINT:PRINT "H A S I L :":PRINT
940   FOR J = 1 TO N
950   PRINT "X(";J;") =";X(J)
960   NEXT J
970   PRINT
980   PRINT"*******************************************************  "
990   RETURN
```